

Wizualizacja danych - Gnuplot

dr hab. Bożena Woźna-Szcześniak

Akademia im. Jan Długosza

bwozna@gmail.com

Laboratorium 9



Plan

- Wiele wykresów w jednym pliku (multiplot)
- Wykres na wykresie
- Wykresy parametryczne
- Wykresy polarne
- Wykresy polarne 3D
- Wektory

Wiele wykresów w jednym pliku

- Aby włączyć tryb przesyłania wielu wykresów do jednego pliku, należy wydać polecenie: `set multiplot`.
- Komenda `set multiplot` użyta w trybie interaktywnym zmienia znak zachęty z `gnuplot>` na `multiplot>`.
- W trybie `multiplot` nie można zmieniać terminala.
- Aby wyłączyć tryb przesyłania wielu wykresów do jednego pliku, należy wydać polecenie: `unset multiplot`.


Składania polecenia `multiplot`

```
set multiplot [ title "{str:title}" ]  
[ layout {int:rows},{int:cols}  
[ rowsfirst | columnsfirst ]  
[ downwards | upwards ]  
[ scale {flt:xfactor}[,{flt:yfactor}] ]  
[ offset {flt:xoff}[,{flt:yoff}] ] ]
```


- Parametr `title`, to tytuł zbioru wykresów.
- Parametr `layout` pozwala na ustalenie rozmiaru tablicy, do której przesyłane będą wykresy.
- Pozostałe parametry mogą być użyte, tylko wtedy, gdy opcja `layout` jest stosowana.

Ustawienie kolejności dodawania wykresów

rowsfirst downwards



columnfirst upwards









Przykładowy skrypt

```
set terminal pdf
set output "w10-1.pdf"
t(x,n) = cos(n*acos(x))
set xrange [-1:1]
set yrange [-1:1]
set multiplot title "Wielomiany Czebyszewa"
    layout 3,2
plot t(x,1)
plot t(x,2)
plot t(x,3)
plot t(x,4)
plot t(x,5)
plot t(x,6)
unset multiplot
```

Wykonanie skryptu z poprzedniego slajdu


Wielomiany Czebyszewa



Osadzanie wykresu w wykresie

```
f(x,a) = x<a ? sqrt(a-x) : 0
g(x,b) = 0.25*x**4 + b*x**2
set multiplot
set grid #rysunek głównego wykresu
plot [-3:2.5][-1:2] f(x,0) notitle w l lw 3
unset grid
unset tics; unset xtics; unset ytics
unset key
set tmargin 0; set bmargin 0
set rmargin 0; set lmargin 0
set size 0.2 #rozmiar małych wykresów
set origin 0.175,0.425 #położenie wykresu
clear # Czysci miejsce dla wykresu
plot [-2.5:2.5][-1.5:2] g(x,-1)
set origin 0.4625,0.125
clear
plot [-2.5:2.5][-0.5:3] g(x,0)
set origin 0.675,0.425
clear
plot [-2.5:2.5][-0.5:3] g(x,1)
unset multiplot
```



Wykonanie skryptu z poprzedniego slajdu



Dwa wykresy obok siebie

```
set terminal pdf
set output "w10-3.pdf"
lgn( x, s ) = exp(-0.5*(log(x)/s)**2)/(x*s*sqrt(2*pi))
clgn( x, s ) = 0.5*(1+erf(log(x)/(s*sqrt(2))))
set multiplot layout 2,1
set bmargin 0 # Wylacza dolny margines
set format x "" # usuwamy etykiety osi x
set ytics 0.2,0.2 # Usuwamy etykiete dla y=0 (start,step)
set key bottom
plot [0:2.5][0:1] clgn(x,0.5) t "s=0.5",
      clgn(x,1) t "s=1.0", clgn(x,5) t "s=5.0"
set bmargin # Wlacza dolny margines
set tmargin 0 # Wylacza górny margines
set format x "%g"
# Usuwamy etykiete dla y=1 (start,step,stop)
set ytics 0,0.2,0.8
set key top
plot [0:2.5][0:1] lgn(x,0.5) t "s=0.5", lgn(x,1) t "s=1.0",
      lgn(x,5) t "s=5.0"
unset multiplot
```

Wykonanie skryptu z poprzedniego slajdu




Wykresy parametryczne

```
set terminal pdf
set output "w10-4.pdf"
plot [-1:1] sqrt(1-x**2), -sqrt(1-x**2)
```


```
set output "w10-4a.pdf"
set parametric
plot [0:2*pi] cos(t), sin(t)
unset parametric
```

Ustawienie kolejności dodawania wykresów

w10-4.pdf



w10-4a.pdf



Wykresy parametryczne

- Dla wykresów parametrycznych definiowane są **trzy** a nie dwa zakresy: `trange`, `xrange`, `yrange`
- Schemat pracy bez i z parametrem:

```
unset parametric
```

```
plot f(x)
```

```
set parametric
```

```
plot t, f(t)
```

- **Przykład:**

```
set parametric
```

```
set output "w10-5.pdf"
```

```
set size square
```


```
r(t) = 1 - exp(-0.25*t/pi)
```

```
# the radius as function of t
```

```
plot [0:25*pi][-1.1:1.1][-1.1:1.1] r(t)*cos(t), r(t)*sin(t)
```

```
unset parametric
```

Wykonanie skryptu z poprzedniego slajdu



Wykresy polarne (biegunowe)

- Na temat współrzędnych biegunowych można poczytać w http://pl.wikipedia.org/wiki/Uk%C5%82ad_wsp%C3%B3%C5%82rz%C4%99dnych_biegunowych.
- Tryb polarny włączamy poleceniem: `set polar`, które nie posiada żadnych innych opcji.
- W trybie polarnym zmienną niezależną jest t i interpretuje się ją jako kąt, a zmienną zależną jest kąt.
- Kąt można ustawiać poleceniem: `set angles [degrees | radians]`
- tryb polarny zdefiniowany jest tylko dla wykresów na płaszczyźnie


polardata.txt

Aby pobrać dane kliknij na [polardata.txt](#) lub na [polardata.txt](#)

Przykład

```
set terminal pdf enhanced
set output "w10-6.pdf"
set polar
set size square
unset border
set format x ""; set format y ""
set xtics scale 0; set ytics scale 0
set grid polar pi/8.0; set xtics 0.25
set tmargin 1; set bmargin 1
set label "0{/Symbol p}" at 1,0 offset char 1,0
set label "+0.5{/Symbol p}" at 0,1 center offset char 0,0.5
set label "-0.5{/Symbol p}" at 0,-1 center offset char 0,-0.5
set label "+/-{/Symbol p}" at -1,0 offset char -5,0
set label "0.5" at 0.5*cos(0.84*pi),0.5*sin(0.84*pi)
    offset char .5,0
set label "1.0" at 1.0*cos(0.83*pi),1.0*sin(0.83*pi)
    offset char .5,0
set key outside left
plot [-2*pi/3:2*pi/3][-1:1][-1:1] "polardata.txt" u 1:2 t
    "Data" w l, 0.27*(2-cos(3*t))+0.75*cos(5*t)) t "Model"
```

Wykonanie skryptu z poprzedniego slajdu




Mapa świata

Aby pobrać dane kliknij na [world.dat](#) lub na [world.dat](#)

```
set terminal pdf enhanced
set output "w10-7.pdf"
plot "world.dat" u 1:2 w l
```

Wykonanie skryptu z poprzedniego slajdu




Mapa świata

Aby pobrać dane kliknij na [world.dat](#) lub na [world.dat](#)

```
set terminal pdf enhanced
set output "w10-7a.pdf"
unset key
set grid
set format ''
set tics scale 0
plot 'world.dat' with lines linestyle 1
```


Wykonanie skryptu z poprzedniego slajdu



Mapa świata 3D - set mapping spherical

```
# AUTHOR: Hagen Wierstorf
set terminal pdf enhanced
set output "w10-8.pdf"
set border lw 1.5
set style line 1 lc rgb '#000000' lt 1 lw 2
set style line 2 lc rgb '#c0c0c0' lt 2 lw 1
unset key; unset border
set tics scale 0
set lmargin screen 0; set bmargin screen 0
set rmargin screen 1; set tmargin screen 1
set format ''
set mapping spherical # tryb biegunowy dla 3D
set angles degrees
set hidden3d ; set xyplane at -1
set view 56,81
set parametric
set isosamples 25
set urange[0:360]; set vrange[-90:90]
r = 0.99
plot r*cos(v)*cos(u),r*cos(v)*sin(u),r*sin(v) w l linestyle 2, \
      'world.dat' with lines linestyle 1
```


Wykonanie skryptu z poprzedniego slajdu




Trójwymiar i wektory

Aby pobrać dane kliknij na [piramida.dat](#) lub na [piramida.dat](#)

```
set terminal pdf enhanced
set output "w10-9.pdf"
unset border
unset tics
unset key
set view 75,35
splot "piramida.dat" i 0 w p pt 7 ps 3, "" i 1
    u 1:2:3:($4-$1):($5-$2):($6-$3) w vectors nohead
```

Wykonanie skryptu z poprzedniego slajdu



Trójwymiar i wektory

Aby pobrać dane kliknij na [piramida.dat](#) lub na [piramida.dat](#)

```
set terminal pdf enhanced
unset border
unset tics
unset key
set view 65,35
set output "w10-9a.pdf"
splot "piramida.dat" i 0 w p pt 6 ps 3, "" i 1
    u 1:2:3:($4-$1):($5-$2):($6-$3)
    w vectors heads filled
```

Wykonanie skryptu z poprzedniego slajdu

